An Introduction to Using Maple in Math 23

The purpose of this document is to help you become familiar with some of the tools that the computer algebra system Maple offers for visualizing solutions to differential equations. I'll start by describing the basics of Maple and then move on to the specific commands that you'll probably find most useful. As you read this, you might find it useful to have Maple running on your computer so that you can try things out as you read about them.

1 General

The command prompt in Maple is the greater than symbol: `>`. Everything entered by you will follow one of these prompts. You'll never have to actually type in the command prompt: Maple does that part for you. After typing in a command, to end and execute it, type a semicolon (`;`) and hit "enter". For example

```
3 + 4;
```

followed by "enter" will ask Maple to add 3 to 4, producing 7 on the following line. You can also end a command with a colon (`:`). The difference is that the colon tells Maple to suppress output. This is useful when you want Maple to perform a computation but when you don't want to see the results right away (or at all). You'll see some instances of its use below.

Addition, subtraction and division all use the usual symbols: `+`,`-`, `/`. Multiplication is done using the asterisk: `*`. Maple *does not* interpret concatenation as multiplication. So, to enter $3y$ you need to type `3*y` or `y*3` and *not* `3y` or `y3`. Exponentiation is achieved by using the caret: `^`. So, for example, to compute 18 multiplied by 2 to the seventh power, you would type

```
18*2^7;
```

followed by "enter".

Overuse of parentheses is never a bad idea when you aren't sure how Maple might choose to associate your operations. For example, if you wanted to enter

$$\frac{4x^2+5}{y-7}$$

you'd type

```
(4*x^2 + 5)/(y - 7)
```

Note that the numerator and denominator are enclosed in parentheses, with the division symbol `/` between them. Any complex fraction should be entered this way, otherwise Maple will get confused and give you funny results. Parentheses should be used in a similar way with exponents: y^{2x-4z} would be entered

```
y^(2*x - 4*z)
```

Such a use of parentheses is necessary any time the exponent consists of more that a single symbol.

Maple also knows most of your basic transcendental functions. It knows how to compute values of the exponential function, natural log, sine, cosine, etc. . Most of these use the usual names you'd see in a math book. For example, to compute $\cos 8$ you'd type `cos(8);` and hit "enter." To find the natural log of $\sqrt{2}$ you'd type `ln(2^(1/2));` and press "enter". Note the exponential notation used for the square root. You can also type `sqrt(2)`. Be aware, however, that to evaluate the exponential function e^x you *do not* simply enter `e^x` as you might expect. You actually type `exp(x)`. So to compute $e^{4/3}$ you would type `exp(4/3);` and hit "enter". If you were to try this out, you'd get the output

$$e^{\left(\frac{4}{3}\right)}$$

which makes it look like Maple didn't do anything! If you want to know the decimal expansion of this or any other "symbolic" number you can use the command `evalf`. So, continuing with the exponential example, typing in `evalf(exp(4/3))` gives the output 3.793667893.

To assign values to variables you use the assignment operator `:=`. So, to create a variable named x and set it equal to 8, you would type

```
x := 8;
```

end hit "enter". To create a variable name *george* and give it the value of x, you'd type

```
george := x;
```

and hit "enter". We'll make use of the assignment operator below.

2 Direction Fields

To plot the direction field for a first order differential equation of the form

$$\frac{dy}{dx} = f(x, y)$$

you can use the command `dfieldplot`. Before you can use this command, however, you must load the `DEtools` package. This is done by typing `with(DEtools):` at the command prompt and pressing enter. Even if you plan on having Maple draw several direction fields, you only need to load the `DEtools` package once.

Let's demonstrate the use of `dfieldplot` with an example. Suppose you want to plot the direction field of the differential equation

$$\frac{dy}{dx} = \frac{x^2 + 5x}{3y - 7y^2}.$$

At the command prompt type

```
dfieldplot(diff(y(x),x) = (x^2 + 5*x)/(3*y(x) - 7*y(x)^2),
                        y(x), x=-5..5, y=-5..5, arrows=line );
```

and hit "enter." The first thing you see here after the word `dfieldplot` is `diff(y(x),x) = (x^2 + 5*x)/(3*y(x) - 7*y(x)^2)`. This specifies the differential equation. The part to the left of the equal sign, `diff(y(x),x)`, is the derivative of $y(x)$ with respect to x, i. e. dy/dx. To the right of the equal sign is `(x^2 + 5*x)/(3*y(x) - 7*y(x)^2)`, which is just the rest of the differential equation. Notice that `y(x)` is used everywhere that y appears in the equation. This is important, otherwise Maple will complain. You can enter the equation however you'd like, as long as you remember that all y's are entered as `y(x)` and dy/dx is entered as `diff(y(x),x)`. So, you could just as well have typed

```
dfieldplot((3*y(x) - 7*y(x)^2)*diff(y(x),x) = x^2 + 5*x,
                       y(x), x=-5..5, y=-5..5, arrows=line );
```

Okay, so what about the rest of the stuff in the `dfieldplot` command? The next thing you see is `y(x)` by itself. This just tells Maple that y is the dependent variable and that it is a function of x. The next two things you see are `x=-5..5` and `y=-5..5`. These just specify the size of the window to display. That is, they indicate the range of x and y values for which the direction field should be plotted. In this example x runs from -5 to 5 and y is in the same range. To change the window, just change the numbers -5 or 5 to whatever you'd like. For example, if you wanted to see what the direction field looked like in the region $0 \leq x \leq 17$, $-3 \leq y \leq -1$ you'd use `x=0..17, y=-3..-1` instead. Finally, the command ends with `arrows=line`. This tells Maple that it should use small line segments (as opposed to arrows) when it draws your direction field.

Should you want to use independent and dependent variables different than x and y, just change them (everywhere that they occur!) to whatever you'd like.

There's one additional option that I didn't include in the example above but that I should mention: it's possible to specify the color of the line segments of the direction field. Not only do colors help fight boredom, but they can be useful if you plan to overlay a plot of a solution curve and a direction field (see below). To tell Maple that you want your direction field to have blue line segments, the `dfieldplot` command should end with

```
...arrows=line, color=blue);
```

You can replace `blue` with any of the following colors: `aquamarine`, `black`, `blue`, `navy`, `coral`, `cyan`, `brown`, `gold`, `green`, `gray`, `grey`, `khaki`, `magenta`, `maroon`, `orange`, `pink`, `plum`, `red`, `sienna`, `tan`, `turquoise`, `violet`, `wheat`, `white`, `yellow`.

3 Solution Curves

There are several ways to plot solution curves of a differential equation using Maple. I'll describe two commands here that plot solutions once you have solved for them by hand. There *are* commands to compute numerical approximations to solutions as well, but I'll wait to describe these until they're needed for our classwork. Also, we will limit ourselves here to the plotting of solutions to *ordinary* differential equations. We'll save the plotting of multivariable functions for later as well.

First, let's talk about the `plot` command. To use it you will need to load the `plots` package by typing `with(plots):` and pressing "enter." As before, this only needs to be done once. The command `plot` will graph any function of the form $y =$ (formula in x). For example, to have Maple draw the graph of $y = x^3 + 2x - 7$ in the region $-3 \leq x \leq 2$ you'd type

```
plot(x^3 + 2*x - 7, x =-3..2 );
```

Notice that you typed `x^3 + 2*x - 7` and not `y = x^3 + 2*x - 7`. The `plot` command does *not* require you to give it the dependent variable, and in fact it will complain if you do so. Also, notice that the the x range is specified as in `dfieldplot` but that the y range is not. Maple adjusts the y range so that the the graph fits in the window. But, if you want to, you can tell Maple the y range you want to see just like you did in the previous section.

The command `plot` will allow you to graph explicit solutions of differential equations, i. e. those solutions that express y explicitly as a function of x. But as you have seen, when you solve some types of equations (e. g. separable equations) our solutions are more naturally expressed implicitly, that is, in the form $f(x, y) = 0$. To plot such solutions you can use the command `implicitplot`. As with `plot`, it can only be used once the `plots` package has been loaded, and this can be done by typing `with(plots):` and pressing "enter." But if you have already loaded the `plots` package, there's no need to do it again.

To see how `implicitplot` works, I'll use an example again. To graph the curve $y^2 + 3y = x^3 - 2x^2 + 1$ in the region $-2 \leq x \leq 3$, $-4 \leq y \leq 2$ you'd type

```
implicitplot(y^2 + 3*y = x^3 -2*x^2 + 1, x=-2..3, y=-4..2);
```

and hit "enter." As with `dfieldplot`, the equation can be entered in any form you like. You could use `y^2 + 3*y - x^3 +2*x^2 = 1` instead if you choose to. Also as before, `x=-2..3, y=-4..2` specifies the window, and can be changed to anything you like.

As with direction fields, you can specify the color of your curves using the `color` option, as described above. Another useful thing you can do is increase the thickness of your curves, so they stand out a little more. This is done by adding the `thickness` option. It's tacked on just like the color option. To set the thickness to 3 and the color to green in the preceding example, you would end the command with

```
...y=-4..2, thickness=3, color=green);
```

The thickness can be any integer ≥ 0. You should try out a few different values to find one that you like.

4 Combining Plots

So now you can plot direction fields and solution curves. But what if you want to put the two together? This is easily done using the `display` command. As with `plot` and `implicitplot`, it can only be used after loading the `plots` package (see above). The easiest way to use it is to compute your plots first, store them as variables and then tell `display` to show them to you.

Let's look at an example. Let's plot the direction field and solutions for

$$\frac{dy}{dx} = \frac{x+2}{y^3+3y-1}.$$

Start by plotting the direction field and storing the result in the variable `p1` (for "plot 1"): type

```
p1 := dfieldplot(diff(y(x),x) = (x+2)/(y(x)^3+3*y(x) - 1), y(x),
                      x=-3..3, y=-3..3, arrows=line, color=black):
```

and hit "enter." Notice that the color of the field has been set to black (so it will contrast with our solution curves) and that the input is ended with a colon. This prevents Maple from actually displaying the plot right now, since you're going to look at it later.

Since the differential equation is separable, you know how to find the solutions (go ahead, it's good practice). They satisfy

$$y^4 + 6y^2 - 4y = 2x^2 + 8x + c.$$

Since this is an implicit equation, you need to use `implicitplot` to plot the curve it represents. Let's do this for two different choices of c, say $c = -1$ and $c = 7$. As with the direction field, store the plots in variables for later use by `display`: type

```
p2 := implicitplot(y^4 + 6*y^2 - 4*y = 2*x^2 + 8*x - 1,
                                          x=-3..3, y=-3..3):
p3 := implicitplot(y^4 + 6*y^2 - 4*y = 2*x^2 + 8*x + 7,
                                          x=-3..3, y=-3..3):
```

and hit "enter" after each line. Again, notice the colon suppressing the output of these computations.

Finally, you can put the three plots together with `display` by typing

```
display(p1,p2,p3);
```

and hitting "enter." If the curves don't stand out enough against the direction field, go back to the `implicitplot` commands you have entered earlier and try changing the `thickness` setting. And if you don't like the view you're getting, go back and change (all of) the x and y ranges.

You can overlay as plots as you would like with `display`. If, for example, you had generated plots `p1`, `p2`, `p3`, `p4`, `p5` and `p6`, you could put them all together by typing

```
display(p1,p2,p3,p4,p5,p6);
```

and pressing "enter".